
1 [bookmark: _Toc122434485][bookmark: _Toc134086713]Overview
This document lists all the essential changes needed to correct problems in the iwd-TTCN3-B2022-09_D23wk12 related to the title of this CR.
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 

2 [bookmark: _Toc134086714]Corrections required
2.1 [bookmark: _Toc134086715][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Function fl_TC_8_1_4_3_2And5_TestBody
	Function name
	fl_TC_8_1_4_3_2And5_TestBody

	Reason for change
	1. According to TS 38.331, 5.3.5.8.3: the UE shall, for each SRB, re-establish the RLC entity for the source PCell. But, SS will not retain the DRBs configuration on source Cell if its SRBs are already released. Therefore, for completeness DRBs should also be released and reconfigured along with the SRBs and PDCP count should also be provided for all again.
2. Due to SRBs release SS would not maintian C-RNTI based contention resolution.
3. Security keys are lost with SRBs and DRB release.
4. FailureInformation message would come on SRB1 so stored PDCP count should be  applied before STEP 4 not after that.
5. f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig returns DRB PDCP count but it is not being stored to apply after STEP 10.
6. At STEP 7 RRCReconfiguration would be sent out after 700 +5 ms as per f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig so STEP 8 T3 configration just after the call doesn’t provide enough time to transmit the rrcReconfiguration message which leads a failure. 

	Summary of change
	1. Stored PDCP count for SRBs and DRB before releasing all and reapplied once configured back including the DRB.
2. Configured CBRA CRNTI Based RACH procedure.
3. Reapplied Security keys.
4. Applied PDCP count on source cell just before the STEP 4 and removed it after STEP 4.
5. Captured the DRB PDCP count value to apply on target cell after STEP 10.
6. Added 800 ms dealy to allow the rrcReconfiguration transmission before applying T3 cell power changes to make the NR Cell 1 off.


	TTCN module
	\ NR5GC\8_1_4\RRC_DAPS_Handover_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function fl_TC_8_1_4_3_2And5_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var SubFrameTiming_Type v_Timing;
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList1; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList2; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList; //@sic R5s221242 sic@
    timer t_Wait := 5.0;
 
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // TC n/a for FR2
                             cs_NR_CellPower(p_TargetCellId, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT2 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT3 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    //Set power level to T0
    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    // Configuration of Target Cell for no RACH response transmission)
    f_NR_SS_RachProcedureConfig(p_TargetCellId, cs_NR_RachProcedureConfig_NoResponse, cs_TimingInfo_Now);

    //@siclog "Step 1" siclog@
    // The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform DAPS handover to NR Cell 2.
    // DRB#n of the first PDU session is configured as DAPS bearer.
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    v_RACH_ConfigGeneric.preambleTransMax := n200; // Table 8.1.4.3.2.3.3-4
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount),
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, v_RRCReconfiguration));
    
    //@siclog "Step 2" siclog@
    // The SS changes the power level setting according to the row "T1".
    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //Acc. 38.331, 5.3.5.8.3: the UE shall, for each SRB, re-establish the RLC entity for the source PCell. When we release an SRB in the SS, the PDCP count is lost, so we need to set it to the original value once the SRB is configured again.
    //@sic R5s221242 sic@
    v_PdcpCountInfoList1 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB1);
    v_PdcpCountInfoList2 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB2);
    v_PdcpCountInfoList := v_PdcpCountInfoList1 & v_PdcpCountInfoList2;
    f_NR_SS_SRBs_DRBs_Release(p_SourceCellId, cs_TimingInfo_Now, {}, tsc_NoCnfReq);
    f_NR_SS_SRBs_DRBs_Config(p_SourceCellId, cs_TimingInfo_Now, omit, tsc_NoCnfReq);

    //@siclog "Step 3-4" siclog@
    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1 within 5s?
    // Check: Does the UE transmit a FailureInformation message with failureType-r16 set to daps-failure on NR Cell 1 within 5s?
    t_Wait.start;
    alt {
      [] t_Wait.timeout{}
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND (p_SourceCellId, cr_38508_RRCReestablishmentRequest(?, ?))){
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRCReestablishmentRequest received on NR Cell 1");
      }
      [] SRB.receive(car_NR_SRB1_RrcPdu_IND (p_SourceCellId, cr_38508_FailureInformation(-, cr_FailureInformation_v1610({daps_failure})))){ //@sic R5s221242 sic@
        t_Wait.stop;
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4"); //@sic R5s221242 sic@
      }
    }
    //@sic R5s221242 sic@
    f_NR_SS_PdcpCount_Set(p_SourceCellId, v_PdcpCountInfoList, cs_TimingInfo_Now);
    
    //@siclog "Step 5" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 5");

    //@siclog "Step 6" siclog@
    // The SS changes the power level setting according to the row "T2".
    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700); // 700 ms
    f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(p_SourceCellId, p_TargetCellId, v_Timing);

    //@siclog "Step 8" siclog@
    // The SS changes the power level setting according to the row "T3" and waits 300ms to ensure T310 expires in NR Cell 1.
    f_NR_SetCellPowerList (v_CellPowerList_AtT3);
    f_Delay(0.3); //@sic R5s221242 sic@
   
    //@siclog "Steps 9-10" siclog@
    // The SS transmits Random Access Response to respond to the latest preamble on NR Cell 2.
    // Check: Does UE transmit an RRCReconfigurationComplete message in NR Cell 2?
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,
                                                             "Step 10",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    //@siclog "Step 11" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB#n of the first PDU session on NR Cell 2?
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 11");

    //@siclog "Steps 12-13" siclog@
    // The SS transmits an RRCReconfiguration message with condition DAPS_HO_ReleaseSource in NR Cell 2.
    // Check: Does the UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, -, false); //@sic R5-230116 sic@
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId);
  }

}



After Change:
	function fl_TC_8_1_4_3_2And5_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList; //WA#WI=1045399
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList;         //WA#WI=1045399
    var NR_SecurityParams_Type v_RRC_SecurityParams;                                      //WA#WI=1045399
    var SubFrameTiming_Type v_Timing;
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList1; //@sic R5s221242 sic@
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList2; //@sic R5s221242 sic@
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList; //@sic R5s221242 sic@
    timer t_Wait := 5.0;
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList; //WA#WI=1045399
    var IntegerList_Type v_DrbIdList ; //WA#WI=1045399
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters; //WA#WI=1045399
    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedure; //WA#WI=1045399
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB; //WA#WI=1045399
      
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // TC n/a for FR2
                             cs_NR_CellPower(p_TargetCellId, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT2 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT3 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    //Set power level to T0
    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    // Configuration of Target Cell for no RACH response transmission)
    f_NR_SS_RachProcedureConfig(p_TargetCellId, cs_NR_RachProcedureConfig_NoResponse, cs_TimingInfo_Now);

    //@siclog "Step 1" siclog@
    // The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform DAPS handover to NR Cell 2.
    // DRB#n of the first PDU session is configured as DAPS bearer.
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    v_RACH_ConfigGeneric.preambleTransMax := n200; // Table 8.1.4.3.2.3.3-4
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount),
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, v_RRCReconfiguration));
    
    //@siclog "Step 2" siclog@
    // The SS changes the power level setting according to the row "T1".
    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //Acc. 38.331, 5.3.5.8.3: the UE shall, for each SRB, re-establish the RLC entity for the source PCell. When we release an SRB in the SS, the PDCP count is lost, so we need to set it to the original value once the SRB is configured again.
    //@sic R5s221242 sic@
    //v_PdcpCountInfoList1 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB1); Not needed
    //v_PdcpCountInfoList2 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB2); Not needed
    //v_PdcpCountInfoList := v_PdcpCountInfoList1 & v_PdcpCountInfoList2;            Not needed

    //WA#WI=1045399 Get PDCP count info for SRBs and DRBs before releasing.
    v_NR_PdcpCountInfoList:=f_NR_SS_PdcpCount_Get(p_SourceCellId);
    v_DrbIdList := f_NR_GetActiveDRBs(p_SourceCellId); //WA#WI=1045399
    
    f_NR_SS_SRBs_DRBs_Release(p_SourceCellId, cs_TimingInfo_Now, v_DrbIdList, tsc_NoCnfReq); //WA#WI=1045399 - release DRB as well.
    // Configure SRBs and DRBs again not just the SRBs.
    v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();//WA#WI=1045399
    v_RadioBearerList := f_NR5GC_GetRadioBearerList(v_DRBInfoList);//WA#WI=1045399
    f_NR_SS_SRBs_DRBs_Config(p_SourceCellId, cs_TimingInfo_Now, v_RadioBearerList, tsc_NoCnfReq);//WA#WI=1045399

    //WA#WI=1045399 - Configure CBRA CRNTI Based RACH procedure again as it has been lost with SRBs release. 
    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_SourceCellId);
    v_NR_RachProcedure := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, -, -, cs_NR_AssignedBWPs_InitialBWP);
    f_NR_SS_RachProcedureConfig(p_SourceCellId, v_NR_RachProcedure);
    
    //WA#WI=1045399 - Reapply security keys.
    v_RRC_SecurityParams := f_NR_Security_Get();
    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_SourceCellId, false, -, v_NR_PdcpCountInfoList); //@sic R5s220123 sic@
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_SourceCellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering
                                            ); // @sic R5s191200_r2 sic@
     
      //WA#WI=1045399 - Apply PDCP count on Cell again so that it can handle messages from UE correctly. i.e., FailureInformation on SRB1
      f_NR_SS_PdcpCount_Set(p_SourceCellId, v_NR_PdcpCountInfoList, cs_TimingInfo_Now);
    //@siclog "Step 3-4" siclog@
    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1 within 5s?
    // Check: Does the UE transmit a FailureInformation message with failureType-r16 set to daps-failure on NR Cell 1 within 5s?
    t_Wait.start;
    alt {
      [] t_Wait.timeout{}
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND (p_SourceCellId, cr_38508_RRCReestablishmentRequest(?, ?))){
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRCReestablishmentRequest received on NR Cell 1");
      }
      [] SRB.receive(car_NR_SRB1_RrcPdu_IND (p_SourceCellId, cr_38508_FailureInformation(-, cr_FailureInformation_v1610({daps_failure})))){ //@sic R5s221242 sic@
        t_Wait.stop;
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4"); //@sic R5s221242 sic@
      }
    }
    //@sic R5s221242 sic@
    
    //@siclog "Step 5" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 5");

    //@siclog "Step 6" siclog@
    // The SS changes the power level setting according to the row "T2".
    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700); // 700 ms
    v_NR_PdcpCountInfoList_DRB:= f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(p_SourceCellId, p_TargetCellId, v_Timing);

    f_Delay(0.8); //WA#WI=1045399 – Allow enough time to transmit rrcReconfiguration message
    //@siclog "Step 8" siclog@
    // The SS changes the power level setting according to the row "T3" and waits 300ms to ensure T310 expires in NR Cell 1.

    f_NR_SetCellPowerList (v_CellPowerList_AtT3);
    f_Delay(0.3); //@sic R5s221242 sic@
   
       
    //@siclog "Steps 9-10" siclog@
    // The SS transmits Random Access Response to respond to the latest preamble on NR Cell 2.
    // Check: Does UE transmit an RRCReconfigurationComplete message in NR Cell 2?
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,
                                                             "Step 10",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));

      
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_DRB, cs_TimingInfo_Now);  //WA#WI=1045399- Set DRB PDCP count. 
    
    
    
    //@siclog "Step 11" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB#n of the first PDU session on NR Cell 2?
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 11");

    //@siclog "Steps 12-13" siclog@
    // The SS transmits an RRCReconfiguration message with condition DAPS_HO_ReleaseSource in NR Cell 2.
    // Check: Does the UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, -, false); //@sic R5-230116 sic@
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId);
  }

}



